Figurate numbers - Figurierte Zahlen

Author: Jochen Ziegenbalg
Email: ziegenbalg.edu@gmail.com
Internet: https://jochen-ziegenbalg.github.io/root/

■ Introduction: Square numbers and triangular numbers

■ Square numbers

■ Construction and recursive description

The most well-known figurate numbers are the *square numbers* (in German: Quadratzahlen), i.e. the numbers 1, 4, 9, 16, 25, 36, 49, 64, 81, 100, 121, ...
They are called square numbers because they can be "arranged" in the shape of squares in an obvious way - and this square arrangement also explains the term "figurate number".



The red angles (i.e. the hook-like shapes) in the diagram were calles "gnomons" in ancient Greek mathematics. Each square is made up by the (blue) previous square plus a (red) gnomon. The numbers belonging to the gnomons of the squares are: 1, 3, 5, 7, 9, ... Since the squares' gnomons start with 1 and, step by step, increase by 2, they are identical to the odd numbers.

As the diagram shows, each square number consists of the previous square plus a suitable gnomon. Or, viewed from the other end, by starting with 1 and adding the next gnomon, we reach the next square, and so on. Since these gnomon numbers obviously are identical to the odd numbers, this shows:

Theorem: Each square number is the sum of consecutive odd numbers (starting with the square number 1).

Theorem (more precise version): Let s be a square number. Then $s = 1 + 3 + 5 + ... + (2 \cdot k + 1)$ for a suitable number k.

Exercise: Describe the relation between s and k in the last theorem.

Let Q_k be the k-th square number $(Q_1 = 1)$. Then, by taking a look at the pattern, we see that obviously the following equations hold

(i) $Q_k = k^2$ (this is called an "explicit" description of Q_k)

(ii) $Q_{k+1} = Q_k + 2 \cdot k + 1$ (this is called a "recursive" description of Q_k)

Exercise: Show that any odd square is congruent to 1 modulo 8.

■ Triangular numbers

The numbers 1, 3, 6, 10, 15, 21, 28, 36, 45, 55, 66, 78, 91, 105, 120, ... are called *triangular numbers* (in German: Dreieckszahlen). They can be represented by using trianglar patterns in the following way:



Let T_k be the k-th triangular number $(T_1 = 1)$. Then the patterns show:

$$T_k = T_{k-1} + k$$

Expanding this equation gives

$$T_k = k + (k-1) + (k-2) + \ldots + 2 + 1 = \sum_{i=1}^{k} i$$

By the argument of the "young Gauß" (Carl Friedrich Gauß, 1777-1855), i.e. by composing triangular "stairs" appropiately,



or more formally by mathematical induction, it follows that

$$T_k = \frac{k \cdot (k+1)}{2}$$

Drawing the triangles (similarly like in Gauss' "stair" visualization above - but without the top blue row), i.e. drawing them with one right angle and two 45-degree angles, gives some insight into the relationship between triangular and square numbers: Each square is the sum of two "adjacent" triangular numbers in the following way.

Theorem: $Q_k = T_k + T_{k-1}$

Proof: Excercise (by a figurate number argument and by mathematical induction).

■ Polygonal numbers

■ Construction and recursive description

Polygonal numbers (triangular numbers, squares, pentagonal numbers, hexagonal numbers, ...) are characterized by two parameters: The number E of vertices (German: Ecken) of the polygon and the stage k at which it is drawn (we will always assume $E \geq 3$ and $k \geq 1$).

By $G[E, k]$ we denote the polygonal number belonging to a polygon with E vertices at stage k. The numbers

$G[3, k]$ are called triangular numbers,
$G[4, k]$ square numbers,
$G[5, k]$ pentagonal numbers,
$G[6, k]$ hexagonal numbers,
$G[7, k]$ heptagonal numbers,
$G[8, k]$ octagonal numbers,
...
$G[E, k]$ E-gonal numbers.

■ Construction of the pattern belonging to the polygonal number $G[E, k]$

Polygonal numbers are the numbers of dots in polygonal patterns in the following way: At stage $k = 1$ every polygonal pattern consits of exaxctly one dot, i.e.: $G[E, 1] = 1$. Let $k \geq 2$. The pattern belonging to $G[E, k]$ evolves out of the pattern belonging to $G[E, k-1]$ by joining an open chain of new dots to $E - 2$ sides of the old pattern so that the vertices make up a new (regular) E-gon with exactly k dots on each of its sides.

In each of the following examples the old pattern is represented by blue dots and the open chain of the new dots is represented by red dots.

Example: The first pentagonal patterns and numbers



1 5 12 22 35

Example: The first hexagonal patterns and numbers



1 6 15 28 45

From this construction the following equation follows at once:

$$G[E,\,k] = G[E,\,k-1] + (E-2)\cdot k - (E-3)$$

Proof: The term $G[E,\,k-1]$ gives the number of dots at stage $k-1$. To this, a chain of dots is added at $E-2$ sides, each side consisting of k dots. This gives $(E-2)\cdot k-(E-3)$ new dots, for the dots at the $(E-3)$ "joins" belong to two sides of the new chain and must not be counted twice.

■ A Mathematica-Program for computing the polygonal number G[E, k]

The following (two-line) *Mathematica* program is a direct implementation of the above given description.

```
G[E_, 1] = 1;
G[E_, k_] := G[E, k - 1] + (E - 2) * k - (E - 3)
```

Next, we consider some uses of this program.

```
G[5, 4]
```

```
22
```

```
Table[G[5, k], {k, 1, 20}]
```

```
{1, 5, 12, 22, 35, 51, 70, 92, 117, 145,
 176, 210, 247, 287, 330, 376, 425, 477, 532, 590}
```

```
TableForm[Table[{k, G[6,  k]}, {k, 1, 20}], TableAlignments → {Right}]
```

1	1
2	6
3	15
4	28
5	45
6	66
7	91
8	120
9	153
10	190
11	231
12	276
13	325
14	378
15	435
16	496
17	561
18	630
19	703
20	780

```
t = Table[G[6,  k], {k, 1, 20}]
```

```
{1, 6, 15, 28, 45, 66, 91, 120, 153, 190,
 231, 276, 325, 378, 435, 496, 561, 630, 703, 780}
```

```
Apply[Plus,  t]
```

```
5530
```

■ Some (empirical) observations

The next table gives the first polygonal numbers from triangles to 10-gons.

```
TableForm[Table[Table[G[E,  k], {k, 1, 18}], {E, 3, 10}],
 TableAlignments -> Right,  TableSpacing → 1]
```

1	3	6	10	15	21	28	36	45	55	66	78	91	105	120	136	153
1	4	9	16	25	36	49	64	81	100	121	144	169	196	225	256	289
1	5	12	22	35	51	70	92	117	145	176	210	247	287	330	376	425
1	6	15	28	45	66	91	120	153	190	231	276	325	378	435	496	561
1	7	18	34	55	81	112	148	189	235	286	342	403	469	540	616	697
1	8	21	40	65	96	133	176	225	280	341	408	481	560	645	736	833
1	9	24	46	75	111	154	204	261	325	396	474	559	651	750	856	969
1	10	27	52	85	126	175	232	297	370	451	540	637	742	855	976	1105

The following program called **`Delta[L_]`** computes the differences of the adjacent numbers in any given list **`L`** of numbers. The program **`Delta[L_, s_]`** iterates this computation of differences **`s`** times.

```
Delta[L_] := Table[L[[i + 1]] - L[[i]], {i, 1, Length[L] - 1}];
Delta[L_,  s_] := If[s == 1, Delta[L],  Delta[Delta[L,  s - 1]]]
```

```
Table[G[6, k], {k, 1, 20}]
```

```
{1, 6, 15, 28, 45, 66, 91, 120, 153, 190,
 231, 276, 325, 378, 435, 496, 561, 630, 703, 780}
```

```
Delta[%]
```

```
{5, 9, 13, 17, 21, 25, 29, 33, 37, 41, 45, 49, 53, 57, 61, 65, 69, 73, 77}
```

```
Delta[%]
```

```
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

Applying the two-parameter **Delta** function from above gives the same values:

```
Delta[Table[G[6, k], {k, 1, 20}] , 2]
```

```
{4, 4, 4, 4, 4, 4, 4, 4, 4, 4, 4, 4, 4, 4, 4, 4, 4, 4}
```

The next program iterates the computation of the differences until all differences are zero.

```
DiffTable[L_] :=
 Module[{T = {L}, L1},
  L1 = Delta[L];
  While[Not[Union[L1] == {0}], T = Append[T, L1]; L1 = Delta[L1]];
  T = Append[T, L1];
  Return[T] ]
```

```
TableForm[
 DiffTable[
  Table[G[6, k], {k, 1, 18}]],
 TableAlignments -> Right, TableSpacing → 1]
```

1	6	15	28	45	66	91	120	153	190	231	276	325	378	435	496	561	630
5	9	13	17	21	25	29	33	37	41	45	49	53	57	61	65	69	
4	4	4	4	4	4	4	4	4	4	4	4	4	4	4	4		
0	0	0	0	0	0	0	0	0	0	0	0	0	0	0			

In the next example, this process is applied to all of the E-gonal numbers with $3 \leq E \leq 10$.

```
TableForm[
 Table[
  DiffTable[
   Table[G[E, k], {k, 1, 18}]], {E, 3, 10}],
 TableSpacing → 2]
```

1	2	1	0
3	3	1	0
6	4	1	0
10	5	1	0
15	6	1	0
21	7	1	0
28	8	1	0
36	9	1	0
45	10	1	0
55	11	1	0
66	12	1	0
78	13	1	0
91	14	1	0
105	15	1	0
120	16	1	0
136	17	1	
153	18		
171			

1	3	2	0
4	5	2	0
9	7	2	0
16	9	2	0
25	11	2	0
36	13	2	0
49	15	2	0
64	17	2	0
81	19	2	0
100	21	2	0
121	23	2	0
144	25	2	0
169	27	2	0
196	29	2	0
225	31	2	0
256	33	2	
289	35		
324			

1	4	3	0
5	7	3	0
12	10	3	0
22	13	3	0
35	16	3	0
51	19	3	0
70	22	3	0
92	25	3	0
117	28	3	0
145	31	3	0
176	34	3	0
210	37	3	0
247	40	3	0
287	43	3	0
330	46	3	0
376	49	3	
425	52		
477			

1	5	4	0
6	9	4	0
15	13	4	0
28	17	4	0
45	21	4	0
66	25	4	0
91	29	4	0
120	33	4	0
153	37	4	0
190	41	4	0
231	45	4	0
276	49	4	0
325	53	4	0
378	57	4	0
435	61	4	0
496	65	4	
561	69		
630			

1	6	5	0
7	11	5	0
18	16	5	0
34	21	5	0
55	26	5	0
81	31	5	0
112	36	5	0
148	41	5	0
189	46	5	0
235	51	5	0
286	56	5	0
342	61	5	0
403	66	5	0
469	71	5	0
540	76	5	0
616	81	5	
697	86		
783			

1	7	6	0
8	13	6	0
21	19	6	0
40	25	6	0
65	31	6	0
96	37	6	0
133	43	6	0
176	49	6	0
225	55	6	0
280	61	6	0
341	67	6	0
408	73	6	0
481	79	6	0
560	85	6	0
645	91	6	0
736	97	6	
833	103		
936			

1	8	7	0
9	15	7	0
24	22	7	0
46	29	7	0
75	36	7	0
111	43	7	0
154	50	7	0
204	57	7	0
261	64	7	0
325	71	7	0
396	78	7	0
474	85	7	0
559	92	7	0
651	99	7	0
750	106	7	0
856	113	7	
969	120		
1089			

1	9	8	0
10	17	8	0
27	25	8	0
52	33	8	0
85	41	8	0
126	49	8	0
175	57	8	0
232	65	8	0
297	73	8	0
370	81	8	0
451	89	8	0
540	97	8	0
637	105	8	0
742	113	8	0
855	121	8	0
976	129	8	
1105	137		
1242			

■ Closed form representations ("formulae")

■ Pyramidal numbers

Pyramidal numbers (tetrahedral numbers, cubes, ...) arise from "stacking" successive polygonal numbers so as to form a pyramid.
The following picture gives a visualisation of the tetrahedral numers.



The next program obviously computes the pyramidal numbers.

```
H[E_, k_] := Sum[G[E, i], {i, 1, k}]
```

An alternative (recursive) description of the pyramidal numbers obviously is given by:

```
H2[E_, 1] = 1;
H2[E_, k_] := H2[E, k - 1] + G[E, k]
```

We compare some results.

```
Table[H[3, k], {k, 1, 22}]
Table[H2[3, k], {k, 1, 22}]
```

```
{1, 4, 10, 20, 35, 56, 84, 120, 165, 220, 286,
 364, 455, 560, 680, 816, 969, 1140, 1330, 1540, 1771, 2024}
```

```
{1, 4, 10, 20, 35, 56, 84, 120, 165, 220, 286,
 364, 455, 560, 680, 816, 969, 1140, 1330, 1540, 1771, 2024}
```

Exercise: In a newspaper article (Sonntag Aktuell, 7. Dez. 1997) it was claimed that the following Christmas tree consits of 3000 champaign glasses. Check the correctness or plausibility of this claim.



■ Sums of trianguar numbers, squares, n-gonal numbers

■ Utilities