

Die Approximation von π

nach Archimedes

(in der Fassung nach Chr. Wolff)

Archimedes von Syrakus (Mathematiker, Physiker): ca. 287-212 v.Chr.

Christian Wolff (Philosoph, Mathematiker): 1679-1754

Autor: Jochen Ziegenbalg

Email: ziegenbalg.edu@gmail.com

Internet: <https://jochen-ziegenbalg.github.io/root/>

■ Literaturhinweise

Beckmann P.: A History of π ; New York 1971

Berggren L., Borwein J., Borwein P.: Pi: A Source Book; Springer Verlag, New York 1997

Ziegenbalg J., Algorithmen von Hammurapi bis Gödel, Heidelberg 1996, 54-59

■ Die Grundidee des Verfahrens von Archimedes

Die Kreislinie wird durch (geradlinig begrenzte) Vielecke angenähert; genauer: Dem Kreis wird jeweils ein regelmäßiges Dreieck, Sechseck, 12-Eck, 24-Eck, 48-Eck, 96-Eck, ... eingeschrieben und umgeschrieben. Durch fortlaufende Erhöhung (Verdopplung) der Eckenzahl nähern sich die jeweiligen Vielecke (Polygone) dem Kreis immer mehr an.

Bezeichnungen:

r: Radius des Ausgangskreises

sn bzw. **s[n]:** Seitenlänge des eingeschriebenen n-Ecks;

sn als *Variable*, **s[n]** als *Funktion*

Sn, S[n]: Seitenlänge für das umgeschriebene n-Eck; Variable, Funktion

un, u[n], Un, U[n]: Umfänge der jeweiligen n-Ecke

In geometrischer Hinsicht wird in der folgenden Entwicklung des Verfahrens ausschliesslich der Satz des Pythagoras, der Strahlensatz und der Satz von der Lage des Schwerpunkts im Dreieck (also die Tatsache, dass er die Seitenhalbierenden im Verhältnis 2 : 1 teilt) verwendet.

■ Festlegung der Anfangswerte für das Dreieck

Hinweis: Satz des Pythagoras, Schwerpunkt im Dreieck (vgl. Ziegenbalg, 1996)

$$s_3 = r \sqrt{3}$$

$$\sqrt{3} \, r$$

$$s_3$$

$$\sqrt{3} \, r$$

$$u_3 = 3 * s_3$$

$$3 \sqrt{3} \, r$$

$$\text{Print}["\text{Abschätzung nach unten (beim Dreieck): } ", \frac{u_3}{2 * r}, ", \dots ", N[\frac{u_3}{2 * r}]]$$

$$\text{Abschätzung nach unten (beim Dreieck): } \frac{3 \sqrt{3}}{2} \dots 2.59808$$

$$S_3 = 2 * r \sqrt{3}$$

$$2 \sqrt{3} \, r$$

$$U_3 = 3 * S_3$$

$$6 \sqrt{3} \, r$$

$$\text{Print}["\text{Abschätzung nach oben (beim Dreieck): } ", \frac{U_3}{2 * r}, ", \dots ", N[\frac{U_3}{2 * r}]]$$

$$\text{Abschätzung nach oben (beim Dreieck): } 3 \sqrt{3} \dots 5.19615$$

■ Rekursionsgleichungen

Wir beziehen uns hier auf die Bezeichnungen und Abbildungen in [Ziegenbalg 1996], Abschnitt 3.2.4, insbesondere Abbildung 3.13.

Nach dem Strahlensatz gilt:

$$\frac{S_n}{s_n} = \frac{r}{r_n} \quad (\text{Arch-1})$$

Nach dem Satz des Pythagoras ist

$$r^2 = r_n^2 + \left(\frac{s_n}{2}\right)^2$$

und somit

$$r_n = r \cdot \sqrt{1 - \left(\frac{s_n}{2 \cdot r}\right)^2} . \quad (\text{Arch-2})$$

Daraus folgt schliesslich

$$S_n = \frac{s_n}{\sqrt{1 - \left(\frac{s_n}{2 \cdot r}\right)^2}} . \quad (\text{Arch-3})$$

Weiterhin gilt nach dem Satz des Pythagoras

$$s_n^2 = \left(\frac{s_n}{2}\right)^2 + d_n^2$$

Für die Strecke d_n gilt mit (Arch-2)

$$d_n = r - r_n = r - r \cdot \sqrt{1 - \left(\frac{s_n}{2 \cdot r}\right)^2} .$$

Somit ist

$$\begin{aligned} s_n^2 &= \left(\frac{s_n}{2}\right)^2 + r^2 \cdot \left(1 - \sqrt{1 - \left(\frac{s_n}{2 \cdot r}\right)^2}\right)^2 \\ &= r^2 \cdot \left(\left(\frac{s_n}{2 \cdot r}\right)^2 + \left(1 - \sqrt{1 - \left(\frac{s_n}{2 \cdot r}\right)^2}\right)^2\right) \\ &= r^2 \cdot \left(\left(\frac{s_n}{2 \cdot r}\right)^2 + \left(1 - 2 \sqrt{1 - \left(\frac{s_n}{2 \cdot r}\right)^2} + \left(1 - \left(\frac{s_n}{2 \cdot r}\right)^2\right)\right)\right) \\ &= r^2 \cdot \left(2 - 2 \sqrt{1 - \left(\frac{s_n}{2 \cdot r}\right)^2}\right) \end{aligned}$$

Daraus folgt schliesslich

$$s_{2n} = r \cdot \sqrt{2 - 2 \sqrt{1 - \left(\frac{s_n}{2 \cdot r}\right)^2}} . \quad (\text{Arch-4})$$

■ Interaktive Berechnung der Daten für das 6-Eck

In der folgenden Zeile wird die Formel (Arch-4) in die eingangs erläuterte Notation umgesetzt. (Die Ergebnisse illustrieren auch die Fähigkeiten von Computeralgebrasystemen zur Umformung und insbesondere Vereinfachung algebraischer Terme - man beachte insbesondere das Ergebnis für s_6 .)

$$s_6 = r \sqrt{2 - 2 \sqrt{1 - \left(\frac{s_3}{2 \cdot r}\right)^2}}$$

r

$$u_6 = 6 * s_6$$

$6 \cdot r$

Nach Formel (Arch-3) ist schliesslich

$$s_6 = \frac{s_6}{\sqrt{1 - \left(\frac{s_6}{2r}\right)^2}}$$

$$\frac{2r}{\sqrt{3}}$$

$$U_6 = 6 * s_6$$

$$4\sqrt{3} r$$

■ Allgemeine Funktionale Beschreibungen

■ Umsetzung des Iterationsschrittes in *Mathematica*

In diesem Abschnitt wird die Computeralgebra-spezifische Umsetzung der Seitenlänge $s[n]$ für einen beliebigen Index n (von der Form $n = 3 \cdot 2^k$) realisiert. Dazu definieren wir zunächst die Funktion `nexts` (für "Nächstes s "), also die Funktion, die s_{2n} aus s_n berechnet. Dies geschieht natürlich mit Hilfe der oben entwickelten Formel (Arch-4).

```
nexts[s_] := r  $\sqrt{2 - 2 \sqrt{1 - \left(\frac{s}{2r}\right)^2}}$  ;

nexts[s3]

r

nexts[s3] == s6

True
```

■ Interaktive Umsetzung der gesamten Iteration

```
nexts[s6]

 $\sqrt{2 - \sqrt{3}} r$ 
```

Dies ist natürlich gleich s_{12} . Wir könnten also s_{12} durch den Funktionswert des Aufrufs `nexts[s3]` definieren und dann mit s_{12} an Stelle von s_6 weitermachen, um s_{24} zu ermitteln, u.s.w. .

```
s12 = nexts[s6]

 $\sqrt{2 - \sqrt{3}} r$ 
```

```
s24 = nexts[s12]
```

$$\sqrt{2 - 2 \sqrt{1 + \frac{1}{4} (-2 + \sqrt{3})}} \quad r$$

Aufgabe: Ermitteln Sie auf diese Weise s48 und s96.

■ Verschachtelung von Funktionen

Die wichtigste Operation mit Funktionen ist die Hintereinanderausführung (Verschachtelung) - gleichermassen ganz allgemein in der Mathematik, wie auch in den Computeralgebrasystem *Mathematica*. Man hätte s24 also auch folgendermassen ermitteln können:

```
nexts[nexts[s6]]
```

$$\sqrt{2 - 2 \sqrt{1 + \frac{1}{4} (-2 + \sqrt{3})}} \quad r$$

oder

```
nexts[nexts[nexts[s3]]]
```

$$\sqrt{2 - 2 \sqrt{1 + \frac{1}{4} (-2 + \sqrt{3})}} \quad r$$

Die *Mathematica*-Funktion `Nest` dient der mehrfachen Verschachtelung von Funktionen. Ihre syntaktische Form ist `Nest[Funktion, Argument, Anzahl]`.

```
Nest[Cos, x, 3]
```

```
Cos[Cos[Cos[x]]]
```

```
Nest[Cos, 1.5, 3]
```

```
0.542405
```

Im folgenden ist noch eine andere mögliche syntaktische Form gegeben:

```
Function[x, Nest[Cos, x, 3]][1.5]
```

```
0.542405
```

Im folgenden ist die Methode von Archimedes zur Kreis-Approximation durch Anwendung der `Nest`-Funktion beschrieben. Man erreicht so eine höhere Stufe der Allgemeinheit des Verfahrens.

Der Aufruf

```
nexts[nexts[nexts[s3]]]
```

$$\sqrt{2 - 2 \sqrt{1 + \frac{1}{4} (-2 + \sqrt{3})}} \, r$$

ist nachdem soeben Gesagten also gleichwertig mit

```
Nest[nexts, s3, 3]
```

$$\sqrt{2 - 2 \sqrt{1 + \frac{1}{4} (-2 + \sqrt{3})}} \, r$$

Wir stehen nun vor dem Problem, die Anzahl der Iterationen von `nexts` allgemein im Falle eines 48-Ecks, 96-Ecks, 192-Ecks, ..., $3 \cdot 2^n$ -Ecks anzugeben.

Wir erstellen zu diesem Zweck eine Tabelle:

Eckenzahl (= n)	3	6	12	24	48	96	...
Anzahl der Iterationen	0	1	2	3	4	5	?
Eckenzahl in der Form $3 \cdot 2^k$	$3 \cdot 2^0$	$3 \cdot 2^1$	$3 \cdot 2^2$	$3 \cdot 2^3$	$3 \cdot 2^4$	$3 \cdot 2^5$	$3 \cdot 2^k$
Zweierlogarithmus von $n/3$	0	1	2	3	4	5	$\text{Log}[2, n/3]$

Eine allgemeine Funktion zur Ermittlung der Seitenlänge des einbeschriebenen n -Ecks ist somit gegeben durch:

$$s[n_] := \text{Nest}[\text{nexts}, s3, \text{Log}[2, \frac{n}{3}]]$$

Ein Beispiel:

```
s[12]
```

$$\sqrt{2 - \sqrt{3}} \, r$$

Die allgemeine Funktion für die Seitenlänge des umbeschriebenen n -Ecks ist nach (Arch-3) gegeben durch:

$$S[n_] := \frac{s[n]}{\sqrt{1 - \left(\frac{s[n]}{2r}\right)^2}}$$

Und schliesslich lauten die Funktionen für die entsprechenden Umfänge:

$$u[n_] := n * s[n]$$

$$U[n_] := n * S[n]$$

■ Demonstration der Symbolverarbeitung

`s[96]`

$$\sqrt{2 - 2 \sqrt{1 + \frac{1}{4} \left(-2 + 2 \sqrt{1 + \frac{1}{4} \left(-2 + 2 \sqrt{1 + \frac{1}{4} (-2 + \sqrt{3})} \right)} \right)}} r$$

`Simplify[s[96]]`

$$\sqrt{2 - \sqrt{2 + \sqrt{2 + \sqrt{2 + \sqrt{3}}}}} r$$

■ Tabelle der Näherungswerte für Pi

```
r = 1.0;
TableForm[
  Table[
    {k, 3 * 2^k,
      PaddedForm[N[ $\frac{u[3 * 2^k]}{2 r}$ ], {12, 10}],
      PaddedForm[N[ $\frac{U[3 * 2^k]}{2 r}$ ], {12, 10}],
      PaddedForm[N[ $\frac{U[3 * 2^k]}{2 r} - \frac{u[3 * 2^k]}{2 r}$ ], {32, 30}, ExponentFunction -> (Null &)]},
    {k, 0, 20}],
  TableSpacing -> {0, 1},
  TableAlignments -> Right,
  TableHeadings -> {{}, {"k", "n", "u", "U", ColumnForm[{"U-u", " "}]}}
```

k	n	u	U	U-u
0	3	2.5980762114	5.1961524227	2.59807621135331400000000000000000
1	6	3.0000000000	3.4641016151	0.46410161513775480000000000000000
2	12	3.1058285412	3.2153903092	0.10956176794322260000000000000000
3	24	3.1326286133	3.1596599421	0.02703132881626313000000000000000
4	48	3.1393502030	3.1460862151	0.00673601208456808800000000000000
5	96	3.1410319509	3.1427145996	0.00168264875485935000000000000000
6	192	3.1414524723	3.1418730500	0.00042057769436176780000000000000
7	384	3.1415576079	3.1416627471	0.00010513914499110650000000000000
8	768	3.1415838921	3.1416101766	0.00002628445637142818000000000000
9	1536	3.1415904632	3.1415970343	0.00000657109347645956600000000000
10	3072	3.1415921060	3.1415937488	0.00000164277208014596000000000000
11	6144	3.1415925166	3.1415929273	0.00000041069294010043220000000000
12	12288	3.1415926186	3.1415927213	0.00000010267322991808210000000000
13	24576	3.1415926453	3.1415926710	0.00000002566830703543133000000000
14	49152	3.1415926453	3.1415926517	0.00000000641707709192473900000000
15	98304	3.1415926453	3.1415926469	0.00000000160426916195888200000000
16	196608	3.1415926453	3.1415926457	0.00000000040106717946741810000000
17	393216	3.1415923038	3.1415923039	0.00000000010026646179994710000000
18	786432	3.1415923038	3.1415923038	0.00000000002506661544998678000000
19	1572864	3.1415868397	3.1415868397	0.00000000000626698692940408400000
20	3145728	3.1415868397	3.1415868397	0.00000000000015667467323510210000

■ Die Werte von Archimedes

Für die numerische Rechnung verwendete Archimedes den Bruch $\frac{265}{153}$ als Näherungswert für $\sqrt{3}$. Wie gut der Näherungswert war, zeigt die folgende kleine Rechnung.

N[$\sqrt{3}$, 20]

1.7320508075688772935

N[265 / 153, 20]

1.7320261437908496732

N[$\sqrt{3} - \frac{265}{153}$, 20]

0.000024663778027620324832

Der von Archimedes gewählte Näherungsbruch weicht also um weniger als 3 Hunderttausendstel $\sqrt{3}$ ab.

Die Berechnung der oben definierten Funktionen $s[n_]$, $S[n_]$, $u[n_]$ und $U[n_]$ hängt vom Anfangswert s_3 ab. Setzt man für s_3 den Archimedischen Anfangswert $\frac{265}{153}$ ein, so kann man hoffen, die Rechnung von Archimedes ein Stück weit zu simulieren. Ganz wird das wohl nicht stimmen, weil Archimedes vermutlich mit den babylonischen Zahlen gerechnet und Zwischenergebnisse möglicherweise anders gerundet hat als *Mathematica* dies tut. In der folgenden Zelle wird diese Simulation nachgespielt. (Damit nach der Simulation wieder der korrekte Wert von s_3 verwendet wird, wird s_3 zum Schluss wieder korrigiert.)

$s_3 = \frac{265}{153};$

s96Archimedes = **s**[96];

S96Archimedes = **S**[96];

u96Archimedes = **u**[96];

U96Archimedes = **U**[96];

s3 = **r** * $\sqrt{3}$;

u96Archimedes / 2

3.14096

U96Archimedes / 2

3.14264

Gelegentlich kann man lesen, dass Archimedes die Grenzen $3 \frac{10}{71}$ und $3 \frac{1}{7}$ für π ermittelt habe. Dies ergäbe numerisch die Werte:

N[$3 + 10 / 71$]

3.14085

N[$3 + 1 / 7$]

3.14286

In der folgenden Tabelle sind die Näherungswerte für π tabellarisch dargestellt. In der ersten Zeile sind die Werte bei (Computeralgebra-) korrektem $\sqrt{3}$, in der zweiten Zeile bei Ersetzung von $\sqrt{3}$ durch $\frac{265}{153}$ und in der dritten Zeile stehen schlicht die Werte $3\frac{10}{71}$, $3\frac{1}{7}$ und ihre Differenz. Man erkennt, dass die Archimedische Methode in der zweiten Zeile sogar bessere Schranken liefert als $3\frac{10}{71}$ und $3\frac{1}{7}$.

```
TableForm[
  {{"korrektes  $\sqrt{3}$  ", u[96] / 2, U[96] / 2, U[96] / 2 - u[96] / 2}, {" $\sqrt{3} = 265/153$  ",
    u96Archimedes / 2, U96Archimedes / 2, U96Archimedes / 2 - u96Archimedes / 2},
  {"[3 10/71 , 3 1/7] ", N[3 + 10 / 71], N[3 + 1 / 7], N[(3 + 1 / 7) - (3 + 10 / 71)]}},
  TableHeadings -> {None, {" ", "einbeschrieben", "umbeschrieben", "Differenz"}}]
```

	einbeschrieben	umbeschrieben	Differenz
korrektes $\sqrt{3}$	3.14103	3.14271	0.00168265
$\sqrt{3} = 265/153$	3.14096	3.14264	0.00168253
[3 10/71 , 3 1/7]	3.14085	3.14286	0.00201207

■ Graphische Darstellungen

■ Hilfsprogramme